

Elementary maths

```

ans = Expand[(a+b)^3] (* mma (ans.101,ans) *)
ans = Factor[-2*x+2*x+a*x-x^2+a*x^2-x^3] (* mma (ans.102,ans) *)
ans = Solve[x^2-4 == 0,x] (* mma (ans.103,ans) *)
ans = Solve[{2*a-b == 3, a+b+c == 1,-b+c == 6},{a,b,c}][[1]] (* mma (ans.104,ans) *)
ans = N[Pi,50] (* mma (ans.105,ans) *)
ans = Apart[1/((1 + x) (5 + x))] (* mma (ans.106,ans) *)
ans = Together[(1/(1 + x) - 1/(5 + x))/4] (* mma (ans.107,ans) *)
ans = Simplify[Tanh[Log[x]]] (* mma (ans.108,ans) *)
ans = Simplify[Tanh[I x]] (* mma (ans.109,ans) *)
ans = Simplify[Sinh[3 x] - 3 Sinh[x] - 4 (Sinh[x])^3] (* mma (ans.110,ans) *)
ans = HoldForm[Tanh[Log[x]]] (* mma (lhs.108,ans) *)
ans = HoldForm[Tanh[I x]] (* mma (lhs.109,ans) *)
ans = HoldForm[Sinh[3 x] - 3 Sinh[x] - 4 (Sinh[x])^3] (* mma (lhs.110,ans) *)

```

```

\begin{align*}
&\&\text{mma}\{ans.101\}\\
&\&\text{mma}\{ans.102\}\\
&\&\text{mma}\{ans.103\}\\
&\&\text{mma}\{ans.104\}\\
&\&\text{mma}\{ans.105\}\\
&\&\text{mma}\{ans.106\}\\
&\&\text{mma}\{ans.107\}\\
\text{mma}\{lhs.108\} &= \text{Mma}\{ans.108\}\\
\text{mma}\{lhs.109\} &= \text{Mma}\{ans.109\}\\
\text{mma}\{lhs.110\} &= \text{Mma}\{ans.110\}
\end{align*}

```

$$ans.101 := a^3 + 3a^2b + 3ab^2 + b^3$$

$$ans.102 := x(x+1)(a-x)$$

$$ans.103 := \{\{x \rightarrow -2\}, \{x \rightarrow 2\}\}$$

$$ans.104 := \left\{a \rightarrow \frac{1}{5}, b \rightarrow -\frac{13}{5}, c \rightarrow \frac{17}{5}\right\}$$

$$ans.105 := 3.1415926535897932384626433832795028841971693993751$$

$$ans.106 := \frac{1}{4(x+1)} - \frac{1}{4(x+5)}$$

$$ans.107 := \frac{1}{(x+1)(x+5)}$$

$$\tanh(\log(x)) = \frac{x^2 - 1}{x^2 + 1} \quad (ans.108)$$

$$\tanh(ix) = i \tan(x) \quad (ans.109)$$

$$\sinh(3x) - 3 \sinh(x) - 4 \sinh^3(x) = 0 \quad (ans.110)$$

Linear Algebra

```
mat = {{2, 3}, {5, 4}}      (* mma (ans.201,mat) *)
ans = Eigenvalues[mat]      (* mma (ans.202,ans) *)
ans = Eigenvectors[mat]     (* mma (ans.203,ans) *)
ans = CharacteristicPolynomial[mat, x] (* mma (ans.204,ans) *)
rhs = {3, 7}
foo = rhs//MatrixForm      (* mma (ans.205,foo) *)
sol = LinearSolve[mat,rhs]  (* mma (ans.206,sol) *)
```

```
\begin{align*}
&\&\text{mma}\{ans.201\}\\
&\&\text{mma}\{ans.202\}\\
&\&\text{mma}\{ans.203\}\\
&\&\text{mma}\{ans.204\}\\
&\&\text{mma}\{ans.205\}\\
&\&\text{mma}\{ans.206\}
\end{align*}
```

$$\begin{aligned}
ans.201 &:= \begin{pmatrix} 2 & 3 \\ 5 & 4 \end{pmatrix} \\
ans.202 &:= \{7, -1\} \\
ans.203 &:= \begin{pmatrix} 3 & 5 \\ -1 & 1 \end{pmatrix} \\
ans.204 &:= x^2 - 6x - 7 \\
ans.205 &:= \begin{pmatrix} 3 \\ 7 \end{pmatrix} \\
ans.206 &:= \left\{ \frac{9}{7}, \frac{1}{7} \right\}
\end{aligned}$$

Limits

```
ans = Limit[Sin[4 x]/x,x->0] (* mma (ans.301,ans) *)
ans = Limit[2^x/x,x->Infinity] (* mma (ans.302,ans) *)
ans = Limit[((x+dx)^2 - x^2)/dx, dx -> 0] (* mma (ans.303,ans) *)
ans = Limit[(4 n + 1)/(3 n - 1),n->Infinity] (* mma (ans.304,ans) *)
ans = Limit[(1+(a/n))^n,n->Infinity] (* mma (ans.305,ans) *)
```

```
\begin{align*}
&\&\text{\mma*{ans.301}}\\
&\&\text{\mma*{ans.302}}\\
&\&\text{\mma*{ans.303}}\\
&\&\text{\mma*{ans.304}}\\
&\&\text{\mma*{ans.305}}
\end{align*}
```

```
ans.301 := 4
ans.302 := ∞
ans.303 := 2x
ans.304 :=  $\frac{4}{3}$ 
ans.305 :=  $e^a$ 
```

Series

```
ans = Series[(1 + x)^(-2), {x, 1, 5}] (* mma (ans.401,ans) *)
ans = Series[Exp[x], {x, 0, 5}] (* mma (ans.402,ans) *)
ans = Sum[1/i^2, {i, 1, 50}] (* mma (ans.403,ans) *)
ans = Sum[1/i^4, {i, 1, Infinity}] (* mma (ans.404,ans) *)
```

```
\begin{align*}
&\&\text{\mma*{ans.401}}\\
&\&\text{\mma*{ans.402}}\\
&\&\text{\mma*{ans.403}}\\
&\&\text{\mma*{ans.404}}
\end{align*}
```

```
ans.401 :=  $\frac{1}{4} - \frac{x-1}{4} + \frac{3}{16}(x-1)^2 - \frac{1}{8}(x-1)^3 + \frac{5}{64}(x-1)^4 - \frac{3}{64}(x-1)^5 + O((x-1)^6)$ 
ans.402 :=  $1 + x + \frac{x^2}{2} + \frac{x^3}{6} + \frac{x^4}{24} + \frac{x^5}{120} + O(x^6)$ 
ans.403 :=  $\frac{3121579929551692678469635660835626209661709}{1920815367859463099600511526151929560192000}$ 
ans.404 :=  $\frac{\pi^4}{90}$ 
```

Calculus

```
ans = D[x Sin[x],x] (* mma (ans.501,ans) *)
ans = D[x Sin[x],x]/.x -> Pi/4 (* mma (ans.502,ans) *)
ans = Integrate[2 Sin[x]^2, {x, a, b}] (* mma (ans.503,ans) *)
ans = Integrate[2 Exp[-x^2],{x, 0, Infinity}] (* mma (ans.504,ans) *)
ans = HoldForm[Integrate[2 Exp[-x^2], {x, 0, Infinity}]] (* mma (lhs.504,ans) *)
ans = Integrate[x^2 + y^2, {x, 0, 1}, {y, 0, x}] (* mma (ans.505,ans) *)
ans = HoldForm[Integrate[x^2 + y^2, {x, 0, 1}, {y, 0, x}]] (* mma (lhs.505,ans) *)
```

```
\begin{align*}
&\&\text{\mma*{ans.501}}\\
&\&\text{\mma*{ans.502}}\\
&\&\text{\mma*{ans.503}}\\
&\text{\mma{lhs.504}}\&=\text{\Mma{ans.504}}\\
&\text{\mma{lhs.505}}\&=\text{\Mma{ans.505}}
\end{align*}
```

$$\text{ans.501} := \sin(x) + x \cos(x)$$

$$\text{ans.502} := \frac{1}{\sqrt{2}} + \frac{\pi}{4\sqrt{2}}$$

$$\text{ans.503} := -a + \sin(a) \cos(a) + b - \sin(b) \cos(b)$$

$$\int_0^{\infty} 2 \exp(-x^2) dx = \sqrt{\pi} \quad (\text{ans.504})$$

$$\int_0^1 \int_0^x (x^2 + y^2) dy dx = \frac{1}{3} \quad (\text{ans.505})$$

Differential equations

```
sol = DSolveValue[y'[x] + y[x] == 2 a Sin[x], y[x], x] (* mma (ans.601,sol) *)
sol = DSolveValue[{y'[x] + y[x] == 2 a Sin[x], y[0] == 0}, y[x], x] (* mma (ans.602,sol) *)
sol = DSolveValue[{y''[x] + y[x] == 0}, y[x], x] (* mma (ans.603,sol) *)
sol = DSolveValue[{y''[x] + y[x] == 0, y[0] == 0, y'[0] == 1}, y[x], x] (* mma (ans.604,sol) *)
ans = DSolveValue[y''[x] + 5 y'[x] - 6 y[x] == 0, y[x], x] (* mma (ans.605,ans) *)
foo = ans /. {C[1] -> 2, C[2] -> 3} (* mma (ans.606,foo) *)
```

```
\begin{align*}
&\&\text{mma}\{ans.601\}\\
&\&\text{mma}\{ans.602\}\\
&\&\text{mma}\{ans.603\}\\
&\&\text{mma}\{ans.604\}\\
&\&\text{mma}\{ans.605\}\\
&\&\text{mma}\{ans.606\}
\end{align*}
```

```
ans.601 := a(sin(x) - cos(x)) + c1 e-x
ans.602 := -a e-x (-ex sin(x) + ex cos(x) - 1)
ans.603 := c2 sin(x) + c1 cos(x)
ans.604 := sin(x)
ans.605 := c1 e-6x + c2 ex
ans.606 := 2 e-6x + 3 ex
```

A table of derivatives and anti-derivatives

This example is based upon a nice example in the Pythontex gallery, see <https://github.com/gpoore/pythontex/>. It uses a tagged block to capture the Mathematica output for later use in the body of the LaTeX table.

```
1  (* Create a list of functions to include in the table *)
2  fun = {Sin[x],      Cos[x],      Tan[x],
3         ArcSin[x],   ArcCos[x],   ArcTan[x],
4         Sinh[x],     Cosh[x],     Tanh[x]};
5
6  eol = {"\\\\" ,      "\\\" ,      "\\\" ,
7         "\\\"[5pt]", "\\\"[5pt]", "\\\"[5pt]",
8         "\\\" ,      "\\\" ,      " "};
9
10 ddxfun = D[#, x] & /@ fun;
11 intfun = Integrate[#, x] & /@ fun;
12
13 ddxfunHold = HoldForm[D[#, x]] & /@ fun;
14 intfunHold = HoldForm[Integrate[#, x]] & /@ fun;
15
16 (* mmaBeg (CalculusTable) *)
17 Do[Print[OutputForm[
18     ToString[TeXForm[ddxfunHold[[i]]]] <> "&=" <>
19     ToString[TeXForm[ddxfun[[i]]]]      <> "\\quad & \\quad"
20     ToString[TeXForm[intfunHold[[i]]]] <> "&=" <>
21     ToString[TeXForm[intfun[[i]]]]      <>
22     eol[[i]]
23     ]], {i,1,9}]
24 (* mmaEnd (CalculusTable) *)
```

```
\begin{align*}
\mma {CalculusTable}
\end{align*}
```

$$\begin{aligned}\frac{\partial \sin(x)}{\partial x} &= \cos(x) \\ \frac{\partial \cos(x)}{\partial x} &= -\sin(x) \\ \frac{\partial \tan(x)}{\partial x} &= \sec^2(x) \\ \frac{\partial \sin^{-1}(x)}{\partial x} &= \frac{1}{\sqrt{1-x^2}} \\ \frac{\partial \cos^{-1}(x)}{\partial x} &= -\frac{1}{\sqrt{1-x^2}} \\ \frac{\partial \tan^{-1}(x)}{\partial x} &= \frac{1}{x^2+1} \\ \frac{\partial \sinh(x)}{\partial x} &= \cosh(x) \\ \frac{\partial \cosh(x)}{\partial x} &= \sinh(x) \\ \frac{\partial \tanh(x)}{\partial x} &= \operatorname{sech}^2(x)\end{aligned}$$

$$\begin{aligned}\int \sin(x) \, dx &= -\cos(x) \\ \int \cos(x) \, dx &= \sin(x) \\ \int \tan(x) \, dx &= -\log(\cos(x)) \\ \int \sin^{-1}(x) \, dx &= \sqrt{1-x^2} + x \sin^{-1}(x) \\ \int \cos^{-1}(x) \, dx &= x \cos^{-1}(x) - \sqrt{1-x^2} \\ \int \tan^{-1}(x) \, dx &= x \tan^{-1}(x) - \frac{1}{2} \log(x^2+1) \\ \int \sinh(x) \, dx &= \cosh(x) \\ \int \cosh(x) \, dx &= \sinh(x) \\ \int \tanh(x) \, dx &= \log(\cosh(x))\end{aligned}$$

Step-by-step integration

This is another nice example drawn from the Pythontex gallery, see <https://github.com/gpoore/pythontex>.

It shows the step-by-step computations of a simple triple integral.

```
1  xmax = 2; ymax = 3; zmax = 4;
2  xmin = 0; ymin = 0; zmin = 0;
3
4  fun = f[x,y,z];
5
6  mytmp = HoldForm[Integrate[#1, {z, #6, #7},
7                      {y, #4, #5},
8                      {x, #2, #3}]] & @@ {fun, xmin, xmax, ymin, ymax, zmin, zmax}; (* mma(lhs.01,mytmp) *)
9
10 fun = x y + y Sin[z] + Cos[x+y];
11
12 myint = HoldForm[Integrate[#1, {z, #6, #7},
13                      {y, #4, #5},
14                      {x, #2, #3}]] & @@ {fun, xmin, xmax, ymin, ymax, zmin, zmax}; (* mma(rhs.01,myint) *)
15
16 myansx = Integrate[#1, {x, #2, #3}] & @@ {fun, xmin, xmax};
17 myintx = HoldForm[Integrate[#1, {z, #4, #5},
18                      {y, #2, #3}]] & @@ {myansx, ymin, ymax, zmin, zmax}; (* mma(rhs.02,myintx) *)
19
20 myansxy = Integrate[#1, {y, #2, #3}] & @@ {myansx, ymin, ymax};
21 myintxy = HoldForm[Integrate[#1, {z, #2, #3}]] & @@ {myansxy, zmin, zmax}; (* mma(rhs.03,myintxy) *)
22
23 myansxyz = Integrate[#1, {z, #2, #3}] & @@ {myansxy, zmin, zmax}; (* mma(rhs.04,myansxyz) *)
24
25 myapprox = N[myansxyz,15]; (* mma(rhs.05,myapprox) *)
```


$$\begin{aligned}
\int_0^4 \int_0^3 \int_0^2 f(x, y, z) dx dy dz &= \int_0^4 \int_0^3 \int_0^2 (xy + \cos(x + y) + y \sin(z)) dx dy dz \\
&= \int_0^4 \int_0^3 2(y + \cos(1 + y) \sin(1) + y \sin(z)) dy dz \\
&= \int_0^4 (8 + \cos(2) + \cos(3) - \cos(5) + 9 \sin(z)) dz \\
&= 41 + 4 \cos(2) + 4 \cos(3) - 9 \cos(4) - 4 \cos(5) \\
&\approx 40.1235865133293
\end{aligned}$$

```

\begin{align*}
\mma{lhs.01} &= \mma{rhs.01} \\
&= \mma{rhs.02} \\
&= \mma{rhs.03} \\
&= \mma{rhs.04} \\
&\approx \mma{rhs.05}
\end{align*}

```

Plotting Bessel functions

This simple example uses Mathematica to produce a plot of the first six Bessel functions. Two plots are shown, one created by Mathematica and a second created by LaTeX using the plotting package `pgfplots` and the data exported from Mathematica.

```
myData = Partition[Flatten[Table[{x, Table[BesselJ[n, x], {n, 0, 5}]}], {x, 0, 15, 0.1}], 7];  
myPlot = Plot[Evaluate[Table[BesselJ[n, x], {n, 0, 5}]], {x, 0, 15}, PlotLegends -> "Expressions"];  
  
Export["example-04-fig.png", myPlot, "PNG"];  
Export["example-04-fig.pdf", myPlot, "PDF"];  
Export["example-04.txt", myData, "Table", "FieldSeparators" -> " "];
```




Figure 1: The first six Bessel functions.

Using pgfplots




Figure 2: The first six Bessel functions.

```
\begin{tikzpicture} % requires \usepackage{pgfplots}
\begin{axis}
[xmin= 0.0, xmax=15.0,
ymin=-0.45, ymax=1.05,
xlabel=$x$, ylabel=$J_n(x)$,
grid=major, grid style={dashed,gray!30},
legend entries = {$J_0$, $J_1$, $J_2$, $J_3$, $J_4$, $J_5$}]
\addplot[blue] table [x index=0, y index=1]{example-04.txt};
\addplot[red] table [x index=0, y index=2]{example-04.txt};
\addplot[green] table [x index=0, y index=3]{example-04.txt};
\addplot[teal] table [x index=0, y index=4]{example-04.txt};
\addplot[orange] table [x index=0, y index=5]{example-04.txt};
\addplot[purple] table [x index=0, y index=6]{example-04.txt};
\end{axis}
\end{tikzpicture}
\captionof{figure}{The first six Bessel functions.} % requires \usepackage{caption}
```

Displaying long expressions

This example uses a simple (though contrived) example of a Taylor series expansion of $1/(1+x)$ to demonstrate the problems that can arise when displaying very long expressions.

```
f[x_] = 1/(1+x)
ans = f[x] (* mma (ans.511,ans) *)
ans = Series[f[x],{x, 0, 10}] (* mma (ans.512,ans) *)
ans = Series[f[x],{x, 0, 20}] (* mma (ans.513,ans) *)
ans = Series[f[x],{x, 0, 23}] (* mma (ans.514,ans) *)
```

```
\begin{dgroup*}[spread={5pt}]
\begin{dmath*} f(x) = \Mma*{ans.511} \end{dmath*}
\begin{dmath*} {} = \Mma*{ans.512} \end{dmath*}
\begin{dmath*} {} = \Mma*{ans.513} \end{dmath*}
\begin{dmath*} {} = \Mma*{ans.514} \end{dmath*}
\begin{dmath*} {} = \Mma*[\hskip 2cm]{ans.514} \end{dmath*}
\end{dgroup*}
```

The first four lines of the following output were set using `\Mma*` while the final line used `\Mma*[\hskip=2cm]`. The last pair of lines displays the output for the same tag `ans.514` and clearly the formatting of the second last line is not ideal as the text has overlapped the tag. This was corrected in the final line by using the option argument `[\hskip=2cm]` in the call to `\Mma*`.

$$f(x) = \frac{1}{x+1} \tag{ans.511}$$

$$= 1 - x + x^2 - x^3 + x^4 - x^5 + x^6 - x^7 + x^8 - x^9 + x^{10} + O(x^{11}) \tag{ans.512}$$

$$= 1 - x + x^2 - x^3 + x^4 - x^5 + x^6 - x^7 + x^8 - x^9 + x^{10} - x^{11} + x^{12} - x^{13} + x^{14} - x^{15} + x^{16} - x^{17} + x^{18} - x^{19} + x^{20} + O(x^{21}) \tag{ans.513}$$

$$= 1 - x + x^2 - x^3 + x^4 - x^5 + x^6 - x^7 + x^8 - x^9 + x^{10} - x^{11} + x^{12} - x^{13} + x^{14} - x^{15} + x^{16} - x^{17} + x^{18} - x^{19} + x^{20} - x^{21} + x^{22} - x^{23} + O(x^{24}) \tag{ans.514}$$

$$= 1 - x + x^2 - x^3 + x^4 - x^5 + x^6 - x^7 + x^8 - x^9 + x^{10} - x^{11} + x^{12} - x^{13} + x^{14} - x^{15} + x^{16} - x^{17} + x^{18} - x^{19} + x^{20} - x^{21} + x^{22} - x^{23} + O(x^{24}) \tag{ans.514}$$

Quadratic convergence of Newton-Raphson iterations

This is a simple example that uses Mathematica to demonstrate the quadratic convergence of Newton-Raphson iterations to the exact root of a non-linear equation.

```
f[x_] = N[x - Exp[-x] , 200]; (* work to 200 decimal digits *)
a = NestList[(# - f[#]/f'[#]) &, SetPrecision[1/2,200], 6]; (* list of x values *)
b = f /@ a; (* list of f values *)
c = #^2 & /@ b; (* list of f^2 values *)
(* mmaBeg(table) *)
Print[OutputForm[
  ToString[0] <> "&" <>
  ToString[NumberForm[a[[1]], 25]] <> "&" <>
  ToString[ScientificForm[b[[1]], 11, NumberFormat -> (SequenceForm[#1, "e", #3] &)] <>
  "\\\\\\\\"
]]
Do[Print[OutputForm[
  ToString[i-1] <> "&" <>
  ToString[NumberForm[a[[i]], 25]] <> "&" <>
  ToString[ScientificForm[b[[i]], 11, NumberFormat -> (SequenceForm[#1, "e", #3] &)] <> "&" <>
  ToString[NumberForm[b[[i]]/c[[i - 1]], 5]] <>
  "\\\\\\\\"
]], {i, 2, 7}]
(* mmaEnd(table) *)
```

Note the clear quadratic convergence in the iterations – the last column settles to approximately -0.11546 independent of the number of iterations. This behaviour would not be seen using normal floating point computations as they are normally limited to no more than 18 decimal digits. This computation used 200 decimal digits.

Newton-Raphson iterations		$x_{n+1} = x_n - f_n/f'_n$, $f(x) = x - e^{-x}$	
n	x_n	$\epsilon_n = x_n - e^{-x_n}$	$\epsilon_n/\epsilon_{n-1}^2$
0	0.50000000000000000000000000000000	-1.0653065971e-1	
1	0.5663110031972181530416492	-1.3045098060e-3	-0.11495
2	0.5671431650348622127865121	-1.9648047172e-7	-0.11546
3	0.5671432904097810286995766	-4.4574262753e-15	-0.11546
4	0.5671432904097838729999687	-2.2941072910e-30	-0.11546
5	0.5671432904097838729999687	-6.0767705445e-61	-0.11546
6	0.5671432904097838729999687	-4.2637434326e-122	-0.11546

```

\def\eps{\epsilong}
\def\RuleA{\vrule depth0pt width0pt height14pt}
\def\RuleB{\vrule depth8pt width0pt height14pt}
\def\RuleC{\vrule depth10pt width0pt height16pt}

\setlength{\tabcolsep}{0.025\textwidth}%

\begin{center}
\begin{tabular}{cccc}%
\halign{\hrule height 1pt}
\multicolumn{4}{c}{\RuleC\rmfamily\bfseries%
Newton-Raphson iterations \quad}

$$x_{n+1} = x_n - f_n/f'_n, \quad f(x) = x - e^{-x}$$

\halign{\hrule height 1pt}
\RuleB $x_n$   $\epsilon_n = x_n - e^{-x_n}$   $\epsilon_n/\epsilon_{n-1}^2$ 
\halign{\hrule height 0.5pt}
\mma{table}
\halign{\hrule height 1pt}
\end{tabular}
\end{center}

```

Using tagged blocks

The following Mathematica code block contains a matched `mmaBeg/mmaEnd` pair, with the tag name `info`, to capture the output from the formatted Mathematica `Print` statements.

```
(* mmaBeg(info) *)
Print[OutputForm["Date : &"    <> ToString[DateString[]]<>"\\\\\\"]]
Print[OutputForm["System : &"  <> ToString[$System]<>"\\\\\\"]]
Print[OutputForm["Version : &" <> ToString[$Version]]]
(* mmaEnd(info) *)
```

```
\bgroup\tt
\begin{tabular}{rl}
  \mma{info}
\end{tabular}
\egroup
```

Here is the output caught from the above block.

```
Date :  Mon 27 Aug 2018 10:16:28
System :  Mac OS X x86 (64-bit)
Version :  11.3.0 for Mac OS X x86 (64-bit) (March 7, 2018)
```

A mixed Mathematica-Python example

This example demonstrates a cooperative effort where Mathematica is used to do the analytic computations while Python is used to plot the data.

The example chosen here is to find and plot the solution to the boundary value problem defined by

$$\frac{d^2y}{dx^2} + 2\frac{dy}{dx} + 10y = 0 \quad \text{with } y(0) = 3, y'(0) = 0$$

This example requires two passes, once for Mathematica and once for Python (and in that order). This example can be run using

```
mmalatex.sh -x -i mixed
pylatex.sh  -x -i mixed
pdflatex    mixed
```

Note that the last pair of commands could also be combined as `pylatex.sh -i mixed`.

The Mathematica code

Here Mathematica is used to first find the general solution of the differential equation. The boundary conditions are then imposed and finally a uniform sampling of the solution is written to a file for later use by Python and Matplotlib.

```
sol = DSolve[y''[x] + 2 y'[x] + 10 y[x] == 0, y, x]
ans = y[x]/.sol[[1]] (* mma(ans.101,ans)*)

sol = DSolve[{y''[x] + 2 y'[x] + 10 y[x] == 0, y[0]==3, y'[0]==0}, y, x]
foo = y[x]/.sol[[1]] (* mma(ans.102,foo)*)
bah = Simplify[y'[x]/.sol[[1]]] (* mma(ans.103,bah)*)

(* now sample y and dy at selected points *)
myData = Table[{x, foo, bah}, {x, 0, 2 Pi, 0.02}];
Export["mixed.txt", myData, "Table", "FieldSeparators" -> " "];
```

The general solution of the differential equation is

$$y(x) = c_1 e^{-x} \sin(3x) + c_2 e^{-x} \cos(3x)$$

while the particular solution satisfying the boundary conditions is given by

$$y(x) = e^{-x} (\sin(3x) + 3 \cos(3x))$$

The Python code

This is a straightforward use of Matplotlib to plot two functions. The code reads the datafile created previously by Mathematica and then calls Matplotlib to plot that data.

```
import numpy as np
import matplotlib.pyplot as plt

plt.matplotlib.rc('text', usetex = True)
plt.matplotlib.rc('grid', linestyle = 'dotted')
plt.matplotlib.rc('figure', figsize = (5.5,4.1)) # (width,height) inches

x, y, dy = np.loadtxt ('mixed.txt', unpack=True)

plt.plot (x,y)
plt.plot (x,dy)

plt.xlim (0.0,4.0)

plt.legend(('y(x)', '$dy(x)/dx$'), loc = 0)
plt.xlabel('$x$')
plt.ylabel('$y(x), \> dy/dx$')
plt.grid(True)
plt.tight_layout(0.5)

plt.savefig('mixed-fig.pdf')
```




Figure 1: The function and its derivative.