Elementary maths

```
ans := expand((a+b)^3):                                # mpl (ans.101,ans)
ans := factor(-2*x+2*x+a*x-x^2+a*x^2-x^3):             # mpl (ans.102,ans)
ans := {solve(x^2-4 = 0,x)}:                           # mpl (ans.103,ans) {...} avoids maple/latex syntax error
sol := solve(x^2-4 = 0,x):                             # multiple roots, can't use mpl(foo,bah) here
ans := {x=sol[1],x=sol[2]}:                            # mpl (ans.104,ans) fixes problem of multiple roots
ans := solve({2*a-b = 3, a+b+c = 1,-b+c = 6},{a,b,c}): # mpl (ans.105,ans)
ans := evalf[50](Pi):                                  # mpl (ans.106,ans)
ans := convert(1/((1 + x)*(5 + x)),parfrac):           # mpl (ans.107,ans)
ans := simplify((1/(1 + x) - 1/(5 + x))/4):            # mpl (ans.108,ans)
ans := simplify(tanh(ln(x))):                          # mpl (ans.109,ans)
ans := simplify(tanh(I*x)):                            # mpl (ans.110,ans)
ans := simplify(sinh(3*x) - 3*sinh(x) - 4*(sinh(x))^3): # mpl (ans.111,ans)
ans := ''tanh(ln(x))'':                                # mpl (lhs.109,ans)
ans := ''tanh(I*x)'':                                  # mpl (lhs.110,ans)
ans := ''sinh(3*x) - 3*sinh(x) - 4*(sinh(x))^3'':      # mpl (lhs.111,ans)
```

$$\begin{aligned}
&\texttt{ans.101} := a^3 + 3\,a^2b + 3\,ab^2 + b^3\\
&\texttt{ans.102} := x\,(x+1)\,(-x+a)\\
&\texttt{ans.103} := \{-2, 2\}\\
&\texttt{ans.104} := \{x=-2, x=2\}\\
&\texttt{ans.105} := \left\{a = \frac{1}{5}, b = -\frac{13}{5}, c = \frac{17}{5}\right\}\\
&\texttt{ans.106} := 3.1415926535897932384626433832795028841971693993751\\
&\texttt{ans.107} := \frac{1}{4\,x+4} - \frac{1}{20+4\,x}\\
&\texttt{ans.108} := \frac{1}{(x+1)\,(5+x)}
\end{aligned}$$

$$\tanh(\ln(x)) = \frac{x^2-1}{x^2+1} \tag{ans.109}$$

$$\tanh(ix) = i\tan(x) \tag{ans.110}$$

$$0 = \sinh(3\,x) - 3\,\sinh(x) - 4\,(\sinh(x))^3 \tag{lhs.111}$$

```latex
\begin{align*}
   &\mpl*{ans.101}\\
   &\mpl*{ans.102}\\
   &\mpl*{ans.103}\\
   &\mpl*{ans.104}\\
   &\mpl*{ans.105}\\
   &\mpl*{ans.106}\\
   &\mpl*{ans.107}\\
   &\mpl*{ans.108}\\
   \mpl{lhs.109} &= \Mpl{ans.109}\\
   \mpl{lhs.110} &= \Mpl{ans.110}\\
   \mpl{ans.111} &= \Mpl{lhs.111}
\end{align*}
```

Linear Algebra

```
with(LinearAlgebra):
mat  := <<2|3>, <5|4>>:                       # mpl (ans.201,mat)
ans  := Eigenvectors(mat,output='list'):
eig1 := ans[1][1]:                            # 1st eigenvalue
eig2 := ans[2][1]:                            # 2nd eigenvalue
v1   := ans[1][3][1]:                         # 1st eigenvector
v2   := ans[2][3][1]:                         # 2nd eigenvector
eig  := <eig1,eig2>:                          # mpl (ans.202,eig)
ans  := <v1|v2>:                              # mpl (ans.203,ans)
ans  := CharacteristicPolynomial(mat,lambda): # mpl (ans.204,ans)
vec  := <3,7>:                                # mpl (ans.205,vec)
sol  := LinearSolve(mat,vec):                 # mpl (ans.206,sol)
```

```
\begin{align*}
   &\mpl*{ans.201}\\
   &\mpl*{ans.202}\\
   &\mpl*{ans.203}\\
   &\mpl*{ans.204}\\
   &\mpl*{ans.205}\\
   &\mpl*{ans.206}
\end{align*}
```

$$\begin{aligned}
&\texttt{ans.201} := \begin{bmatrix} 2 & 3 \\ 5 & 4 \end{bmatrix}\\
&\texttt{ans.202} := \begin{bmatrix} -1 \\ 7 \end{bmatrix}\\
&\texttt{ans.203} := \begin{bmatrix} -1 & \frac{3}{5} \\ 1 & 1 \end{bmatrix}\\
&\texttt{ans.204} := \lambda^2 - 6\,\lambda - 7\\
&\texttt{ans.205} := \begin{bmatrix} 3 \\ 7 \end{bmatrix}\\
&\texttt{ans.206} := \begin{bmatrix} \frac{9}{7} \\ \frac{1}{7} \end{bmatrix}
\end{aligned}$$

Limits

```
ans := limit(sin(4*x)/x,x=0):                    # mpl (ans.301,ans)
ans := limit(2^x/x,x=infinity):                  # mpl (ans.302,ans)
ans := limit(((x+dx)^2 - x^2)/dx, dx=0):         # mpl (ans.303,ans)
ans := limit((4*n + 1)/(3*n - 1),n=infinity):    # mpl (ans.304,ans)
ans := limit((1+(a/n))^n,n=infinity):            # mpl (ans.305,ans)
```

```
\begin{align*}
   &\mpl*{ans.301}\\
   &\mpl*{ans.302}\\
   &\mpl*{ans.303}\\
   &\mpl*{ans.304}\\
   &\mpl*{ans.305}
\end{align*}
```

$$\begin{aligned}
&\texttt{ans.301} := 4\\
&\texttt{ans.302} := \infty\\
&\texttt{ans.303} := 2\,x\\
&\texttt{ans.304} := \frac{4}{3}\\
&\texttt{ans.305} := \mathrm{e}^{a}
\end{aligned}$$

Series

```
ans := series((1 + x)^(-2), x=1, 6):           # mpl (ans.401,ans)
ans := series(exp(x), x=0, 6):                 # mpl (ans.402,ans)
ans := sum(1/n^2, n=1..50):                    # mpl (ans.403,ans)
ans := sum(1/n^4, n=1..infinity):              # mpl (ans.404,ans)
```

```
\begin{align*}
   &\mpl*{ans.401}\\
   &\mpl*{ans.402}\\
   &\mpl*{ans.403}\\
   &\mpl*{ans.404}
\end{align*}
```

$$\begin{aligned}
&\texttt{ans.401} := (\frac{1}{4} - \frac{1}{4}\,(x-1) + \frac{3}{16}\,(x-1)^{2} - \frac{1}{8}\,(x-1)^{3} + \frac{5}{64}\,(x-1)^{4} - \frac{3}{64}\,(x-1)^{5} + O\left((x-1)^{6}\right))\\
&\texttt{ans.402} := (1 + x + \frac{1}{2}x^{2} + \frac{1}{6}x^{3} + \frac{1}{24}x^{4} + \frac{1}{120}x^{5} + O\left(x^{6}\right))\\
&\texttt{ans.403} := \frac{3121579929551692678469635660835626209661709}{1920815367859463099600511526151929560192000}\\
&\texttt{ans.404} := \frac{\pi^{4}}{90}
\end{aligned}$$

Calculus

```
ans := diff(x*sin(x),x):                          # mpl (ans.501,ans)
ans := eval(diff(x*sin(x),x),x=Pi/4):             # mpl (ans.502,ans)
ans := int(2*sin(x)^2, x=a..b):                   # mpl (ans.503,ans)
ans := int(2*exp(-x^2),x=0..infinity):            # mpl (ans.504,ans)
ans := ''int(2*exp(-x^2),x=0..infinity)'':        # mpl (lhs.504,ans)
ans := int(int(x^2 + y^2,  y=0..x),x=0..1):       # mpl (ans.505,ans)
ans := ''int(int(x^2 + y^2,  y=0..x),x=0..1)'':   # mpl (lhs.505,ans)
```

```
\begin{align*}
   &\mpl*{ans.501}\\
   &\mpl*{ans.502}\\
   &\mpl*{ans.503}\\
   \mpl{lhs.504}&=\Mpl{ans.504}\\
   \mpl{lhs.505}&=\Mpl{ans.505}
\end{align*}
```

$$\texttt{ans.501} := \sin(x) + x\cos(x)$$

$$\texttt{ans.502} := \frac{\sqrt{2}}{2} + \frac{\pi\sqrt{2}}{8}$$

$$\texttt{ans.503} := \sin(a)\cos(a) - a - \sin(b)\cos(b) + b$$

$$\int_0^\infty 2\,\mathrm{e}^{-x^2}\,\mathrm{d}x = \sqrt{\pi} \tag{ans.504}$$

$$\int_0^1\int_0^x x^2 + y^2\,\mathrm{d}y\,\mathrm{d}x = \frac{1}{3} \tag{ans.505}$$

Differential equations

```
ode := diff(y(x),x) + y(x) = 2*a*sin(x):
ics := y(0) = 0:
ans := rhs(dsolve(ode)):                      # mpl (ans.601,ans)
ans := rhs(dsolve([ics,ode])):                # mpl (ans.602,ans)

ode := diff(y(x),x,x) + y(x) = 0:
ics := y(0)=0, (D(y))(0) = 1:
ans := rhs(dsolve(ode)):                      # mpl (ans.603,ans)
ans := rhs(dsolve([ics,ode])):                # mpl (ans.604,ans)

ode := diff(y(x),x,x) + 5*diff(y(x),x) - 6*y(x) = 0:
ans := rhs(dsolve(ode)):                      # mpl (ans.605,ans)
sol := eval(ans,[_C1=2,_C2=3]):               # mpl (ans.606,sol)
```

```
\begin{align*}
   &\mpl*{ans.601}\\
   &\mpl*{ans.602}\\
   &\mpl*{ans.603}\\
   &\mpl*{ans.604}\\
   &\mpl*{ans.605}\\
   &\mpl*{ans.606}
\end{align*}
```

$$
\begin{aligned}
\texttt{ans.601} &:= -\cos(x)\,a + a\sin(x) + \mathrm{e}^{-x}C_1\\
\texttt{ans.602} &:= -\cos(x)\,a + a\sin(x) + \mathrm{e}^{-x}a\\
\texttt{ans.603} &:= C_1\,\sin(x) + C_2\,\cos(x)\\
\texttt{ans.604} &:= \sin(x)\\
\texttt{ans.605} &:= C_1\,\mathrm{e}^{-6\,x} + C_2\,\mathrm{e}^{x}\\
\texttt{ans.606} &:= 2\,\mathrm{e}^{-6\,x} + 3\,\mathrm{e}^{x}
\end{aligned}
$$

A table of derivatives and anti-derivatives

This example is based upon a nice example in the Pythontex gallery, see https://github.com/gpoore/pythontex/. It uses a tagged block to capture the Maple output for later use in the body of the LaTeX table.

```
# Create a list of functions to include in the table
funcs := [[sin(x),"\\\\"],         [cos(x),"\\\\"],         [tan(x),"\\\\"],
          [arcsin(x),"\\\\[5pt]"], [arccos(x),"\\\\[5pt]"], [arctan(x),"\\\\[5pt]"],
          [sinh(x),"\\\\"],        [cosh(x),"\\\\"],        [tanh(x)," "]]:

# mplBeg (CalculusTable)
for foo in funcs do
    func := foo[1]:
    eol  := foo[2]:
    myddx := ''diff''(func,x):
    myint := ''int''(func,x):
    Print(cat(Latex(myddx),"&=",Latex(diff(func,x)),"\\quad & \\quad")):
    Print(cat(Latex(myint),"&=",Latex(int(func,x)),eol)):
end do:
# mplEnd (CalculusTable)
```

```
\begin{align*}
   \mpl {CalculusTable}
\end{align*}
```

$$\frac{\mathrm{d}}{\mathrm{d}x}\sin\left(x\right) = \cos\left(x\right)$$

$$\frac{\mathrm{d}}{\mathrm{d}x}\cos\left(x\right) = -\sin\left(x\right)$$

$$\frac{\mathrm{d}}{\mathrm{d}x}\tan\left(x\right) = 1 + \left(\tan\left(x\right)\right)^{2}$$

$$\frac{\mathrm{d}}{\mathrm{d}x}\arcsin\left(x\right) = \frac{1}{\sqrt{-x^{2} + 1}}$$

$$\frac{\mathrm{d}}{\mathrm{d}x}\arccos\left(x\right) = -\frac{1}{\sqrt{-x^{2} + 1}}$$

$$\frac{\mathrm{d}}{\mathrm{d}x}\arctan\left(x\right) = \left(x^{2} + 1\right)^{-1}$$

$$\frac{\mathrm{d}}{\mathrm{d}x}\sinh\left(x\right) = \cosh\left(x\right)$$

$$\frac{\mathrm{d}}{\mathrm{d}x}\cosh\left(x\right) = \sinh\left(x\right)$$

$$\frac{\mathrm{d}}{\mathrm{d}x}\tanh\left(x\right) = 1 - \left(\tanh\left(x\right)\right)^{2}$$

$$\int\sin\left(x\right)\;\mathrm{d}x = -\cos\left(x\right)$$

$$\int\cos\left(x\right)\;\mathrm{d}x = \sin\left(x\right)$$

$$\int\tan\left(x\right)\;\mathrm{d}x = -\ln\left(\cos\left(x\right)\right)$$

$$\int\arcsin\left(x\right)\;\mathrm{d}x = x\arcsin\left(x\right) + \sqrt{-x^{2} + 1}$$

$$\int\arccos\left(x\right)\;\mathrm{d}x = x\arccos\left(x\right) - \sqrt{-x^{2} + 1}$$

$$\int\arctan\left(x\right)\;\mathrm{d}x = x\arctan\left(x\right) - \frac{\ln\left(x^{2} + 1\right)}{2}$$

$$\int\sinh\left(x\right)\;\mathrm{d}x = \cosh\left(x\right)$$

$$\int\cosh\left(x\right)\;\mathrm{d}x = \sinh\left(x\right)$$

$$\int\tanh\left(x\right)\;\mathrm{d}x = \ln\left(\cosh\left(x\right)\right)$$

Step-by-step integration

This is another nice example drawn from the Pythontex gallery, see https://github.com/gpoore/pythontex.

It shows the step-by-step computations of a simple triple integral.

```
# Define limits of integration
x_max := 2:   y_max := 3:   z_max := 4:
x_min := 0:   y_min := 0:   z_min := 0:

ans := int(f(x,y,z), [x=x_min..x_max, y=y_min..y_max, z=z_min..z_max]):           # mpl(lhs.01,ans)

f := (x,y,z) -> x*y + y*sin(z) + cos(x+y):

ans := ''int''(''int''(''int''(f(x,y,z), x=x_min..x_max), y=y_min..y_max), z=z_min..z_max):    # mpl(rhs.01,ans)
ans := ''int''(''int''(int(f(x,y,z), x=x_min..x_max), y=y_min..y_max), z=z_min..z_max):        # mpl(rhs.02,ans)
ans := ''int''(int(int(f(x,y,z), x=x_min..x_max), y=y_min..y_max), z=z_min..z_max):            # mpl(rhs.03,ans)
ans := int(int(int(f(x,y,z), x=x_min..x_max), y=y_min..y_max), z=z_min..z_max):                # mpl(rhs.04,ans)

# And now, a numerical approximation
ans := evalf[15](ans):                                                              # mpl(rhs.05,ans)
```

$$
\begin{aligned}
\int_0^4\int_0^3\int_0^2 f(x,y,z)\,\mathrm{d}x\,\mathrm{d}y\,\mathrm{d}z &= \int_0^4\int_0^3\int_0^2 yx + y\sin(z) + \cos(x+y)\,\mathrm{d}x\,\mathrm{d}y\,\mathrm{d}z\\
&= \int_0^4\int_0^3 -\sin(y) + 2\,y + 2\,y\sin(z) + \sin(2+y)\,\mathrm{d}y\,\mathrm{d}z\\
&= \int_0^4 8 + \cos(2) + 9\,\sin(z) + \cos(3) - \cos(5)\,\mathrm{d}z\\
&= 41 + 4\,\cos(2) - 9\,\cos(4) + 4\,\cos(3) - 4\,\cos(5)\\
&\approx 40.1235865133292
\end{aligned}
$$

```
\begin{align*}
   \mpl{lhs.01} &= \mpl{rhs.01}\\
                &= \mpl{rhs.02}\\
                &= \mpl{rhs.03}\\
                &= \mpl{rhs.04}\\
                &\approx \mpl{rhs.05}
\end{align*}
```

Plotting Bessel functions

This simple example uses Maple to produce a plot of the first six Bessel functions. Two plots are shown, one created by Maple and a second created by LaTeX using the plotting package `pgfplots` and the data exported from Maple.

```
with(plottools):

myPlot := plot([seq(BesselJ(i, z), i = 0 .. 5)], z = 0 .. 15):

exportplot("example-04-fig.jpeg",myPlot,"JPEG"):   # Maple18

a,b,n := 0.0,15.0,150:            # domain and number of samples
dx := (b-a)/n:                    # uniform step

fd := fopen ("example-04.txt", WRITE):
for i from 0 to n by 1 do
   x := a + dx*i:
   fprintf(fd,"% .10e % .10e % .10e % .10e % .10e % .10e % .10e\n",x,
              seq(evalf(BesselJ(k,x)),k=0..5)):
end do:
fclose(fd):
```



Figure 1: The first six Bessel functions.

Using pgfplots



Figure 2: The first six Bessel functions.

```
\begin{tikzpicture} % requires \usepackage{pgfplots}
   \begin{axis}
      [xmin= 0.0,  xmax=15.0,
       ymin=-0.45, ymax=1.05,
       xlabel=$x$, ylabel=$J_n(x)$,
       grid=major, grid style={dashed,gray!30},
       legend entries = {$J_0$, $J_1$, $J_2$, $J_3$, $J_4$, $J_5$}]
       \addplot[blue]   table [x index=0, y index=1]{example-04.txt};
       \addplot[red]    table [x index=0, y index=2]{example-04.txt};
       \addplot[green]  table [x index=0, y index=3]{example-04.txt};
       \addplot[teal]   table [x index=0, y index=4]{example-04.txt};
       \addplot[orange] table [x index=0, y index=5]{example-04.txt};
       \addplot[purple] table [x index=0, y index=6]{example-04.txt};
   \end{axis}
\end{tikzpicture}
\captionof{figure}{The first six Bessel functions.} % requires \usepackage{caption}
```

Displaying long expressions

This example uses a simple (though contrived) example of a Taylor series expansion of $1/(1+x)$ to demonstrate the problems that can arise when displaying very long expressions.

```
f   := x -> 1/(1+x):
ans := f(x):                      # mpl (ans.511,ans)
ans := series(f(x),x=0, 10):      # mpl (ans.512,ans)
ans := series(f(x),x=0, 20):      # mpl (ans.513,ans)
ans := series(f(x),x=0, 23):      # mpl (ans.514,ans)
```

```
\begin{dgroup*}[spread={5pt}]
   \begin{dmath*} f(x) = \Mpl*{ans.511} \end{dmath*}
   \begin{dmath*}    {}= \Mpl*{ans.512} \end{dmath*}
   \begin{dmath*}    {}= \Mpl*{ans.513} \end{dmath*}
   \begin{dmath*}    {}= \Mpl*{ans.514} \end{dmath*}
   \begin{dmath*}    {}= \Mpl*[\hskip 2cm]{ans.514} \end{dmath*}
\end{dgroup*}
```

The first four lines of the following output were set using `\Mpl*` while the final line used `\Mpl*[\hskip=2cm]`. The last pair of lines displays the output for the same tag `ans.514` and clearly the formatting of the second last line is not ideal as the text has overlapped the tag. This was corrected in the final line by using the option argument `[\hskip=2cm]` in the call to `\Mpl*`.

$$f(x) = (1+x)^{-1} \qquad (\texttt{ans.511})$$

$$= \left(1 - x + x^2 - x^3 + x^4 - x^5 + x^6 - x^7 + x^8 - x^9 + O\left(x^{10}\right)\right) \qquad (\texttt{ans.512})$$

$$= \left(1 - x + x^2 - x^3 + x^4 - x^5 + x^6 - x^7 + x^8 - x^9 + x^{10} - x^{11} + x^{12} - x^{13} + x^{14} - x^{15} + x^{16} - x^{17} + x^{18} - x^{19} + O\left(x^{20}\right)\right) \qquad (\texttt{ans.513})$$

$$= \left(1 - x + x^2 - x^3 + x^4 - x^5 + x^6 - x^7 + x^8 - x^9 + x^{10} - x^{11} + x^{12} - x^{13} + x^{14} - x^{15} + x^{16} - x^{17} + x^{18} - x^{19} + x^{20} - x^{21} + x^{22} + O\left(x^{23}\right)\right) \qquad (\texttt{ans.514})$$

$$= \left(1 - x + x^2 - x^3 + x^4 - x^5 + x^6 - x^7 + x^8 - x^9 + x^{10} - x^{11} + x^{12} - x^{13} + x^{14} - x^{15} + x^{16} - x^{17} + x^{18} - x^{19} + x^{20} - x^{21} + x^{22} + O\left(x^{23}\right)\right) \qquad (\texttt{ans.514})$$

Quadratic convergence of Newton-Raphson iterations

This is a simple example that uses Maple to demonstrate the quadratic converegnce of Newton-Raphson iterations to the exact root of a non-linear equation.

```
# mplBeg (table)

   Digits := 200:

    f := x -> x - exp(-x):
   df := x -> eval (diff (f(u),u), u=x):

   x_new := 0.5:
   f_new := f (x_new):

   Print (sprintf("\\RuleA % 2d &% .25f &% .10e &\\\\",
                  0,x_new,f_new)):

   for n from 1 to 6 do
      x_old := x_new:
      x_new := x_old - f(x_old)/df(x_old):
      f_old := evalf (f (x_old)):
      f_new := evalf (f (x_new)):
      ratio := evalf (f_new / f_old^2):
      Print (sprintf("\\RuleA % 2d &% .25f &% .10e &% .5f\\\\",
                     n,x_new,f_new,ratio)):
   end do:

# mplEnd (table)
```

```
\def\eps{\epsilon}
\def\RuleA{\vrule depth0pt  width0pt height14pt}
\def\RuleB{\vrule depth8pt  width0pt height14pt}
\def\RuleC{\vrule depth10pt width0pt height16pt}

\setlength{\tabcolsep}{0.025\textwidth}%

\begin{center}
\begin{tabular}{cccc}%
   \noalign{\hrule height 1pt}
   \multicolumn{4}{c}{\RuleC\rmfamily\bfseries%
   Newton-Raphson iterations \quad%
   $x_{n+1} = x_n - f_n/f'_n\ ,\quad f(x) = x-e^{-x}$}\\
   \noalign{\hrule height 1pt}
   \RuleB$n$&$x_n$&
         $\eps_{n} = x_{n} - e^{-x_{n}}$&
         $\eps_{n}/\eps_{n-1}^2$\\
   \noalign{\hrule height 0.5pt}
   \mpl{table}
   \noalign{\hrule height 1pt}
\end{tabular}
\end{center}
```

Note the clear quadratic convergence in the iterations – the last column settles to approximately -0.11546 independent of the number of iterations. This behaviour would not be seen using normal floating point computations as they are normally limited to no more than 18 decimal digits. This computation used 200 decimal digits.

Newton-Raphson iterations $x_{n+1} = x_n - f_n/f'_n$, $f(x) = x - e^{-x}$			
n	x_n	$\epsilon_n = x_n - e^{-x_n}$	$\epsilon_n/\epsilon_{n-1}^2$
0	0.5000000000000000000000000	-1.0653065971e-01	
1	0.5663110031972181530416492	-1.3045098060e-03	-0.11495
2	0.5671431650348622127865121	-1.9648047172e-07	-0.11546
3	0.5671432904097810286995766	-4.4574262753e-15	-0.11546
4	0.5671432904097838729999687	-2.2941072910e-30	-0.11546
5	0.5671432904097838729999687	-6.0767705445e-61	-0.11546
6	0.5671432904097838729999687	-4.2637434326e-122	-0.11546

Using tagged blocks

The following Maple code block contains a matched `mplBeg/mplEnd` pair, with the tag name `info`, to capture the output from the formatted Maple `Print` statements.

```
# mplBeg(info)
Print(cat("date :  &",FormatTime("%a %d %b %Y %H:%M:%S"),"\\\\")):
Print(cat("pwd :   &",getenv("PWD"),"\\\\")):
Print(cat("shell : &",getenv("SHELL"),"\\\\")):
Print(cat("home :  &",getenv("HOME"))):
# mplEnd(info)
```

```
\bgroup\tt
\begin{tabular}{rl}
   \mpl{info}
\end{tabular}
\egroup
```

Here is the output caught from the above block.

date :	Wed 17 Nov 2021 13:54:13
pwd :	/Users/leo/GitHub/leo-brewin/hybrid-latex/maple/examples
shell :	/usr/local/bin/bash
home :	/Users/leo

A mixed Maple-Python example

This example demonstrates a cooperative effort where Maple is used to do the analytic computations while Python is used to plot the data.

The example chosen here is to find and plot the solution to the boundary value problem defined by

$$\frac{d^2y}{dx^2} + 2\frac{dy}{dx} + 10y = 0 \qquad \text{with } y(0) = 3,\ y'(0) = 0$$

This example requires two passes, once for Maple and once for Python (and in that order). This example can be run using

```
mpllatex.sh -x -i mixed
pylatex.sh  -x -i mixed
pdflatex          mixed
```

Note that the last pair of commands could also be combined as `pylatex.sh -i mixed`.

The Maple code

Here Maple is used to first find the general solution of th differential equation. The boundary conitions are then imposed and finally a uniform sampling of the solution is written to a file for later use by Python and Matplotlib.

```
# a second order ode
ode := diff(y(x), x, x) + 2*diff(y(x),x) + 10*y(x) = 0:  # mpl (ans.101,ode)

# find the general solution
ans := dsolve(ode):                          # mpl (ans.102,ans)

# set initial conditions
ics := y(0) = 3, (D(y))(0) = 0:
tmp := {ics}:                                # mpl (ans.103,tmp)

# find the particular solution
 f := rhs(dsolve([ics, ode])):               # mpl (ans.104,f)
df := diff(f,x):                             # mpl (ans.105,df)

 y := x -> f:
dy := x -> df:

# now sample y and dy at selected points
a,b,n := 0.0,2.0*Pi,300:        # domain and number of samples
dx := (b-a)/n:                  # uniform step

fd := fopen ("mixed.txt", WRITE):
for i from 0 to n by 1 do
   x := a + dx*i:
   fprintf(fd,"% .10e % .10e % .10e\n",x,evalf(y(x)),evalf(dy(x))):
end do:
fclose(fd):
```

The general solution of the differential equation is

$$y\,(x) = C_1\,\mathrm{e}^{-x}\sin\,(3\,x) + C_2\,\mathrm{e}^{-x}\cos\,(3\,x)$$

while the particular solution satifying the boundary conditions is given by

$$y(x) = \mathrm{e}^{-x}\sin\,(3\,x) + 3\,\mathrm{e}^{-x}\cos\,(3\,x)$$

The Python code

This is a straighforward use of Matplotlib to plot two functions. The code reads the datafile created previously by Maple and then calls Matplotlib to plot that data.

```
import numpy as np
import matplotlib.pyplot as plt

plt.matplotlib.rc('text', usetex = True)
plt.matplotlib.rc('grid', linestyle = 'dotted')
plt.matplotlib.rc('figure', figsize = (5.5,4.1)) # (width,height) inches

x, y, dy = np.loadtxt ('mixed.txt', unpack=True)

plt.plot (x,y)
plt.plot (x,dy)

plt.xlim (0.0,4.0)

plt.legend(('$y(x)$', '$dy(x)/dx$'), loc = 0)
plt.xlabel('$x$')
plt.ylabel('$y(x),\> dy/dx$')
plt.grid(True)
plt.tight_layout(0.5)

plt.savefig('mixed-fig.pdf')
```



Figure 1: The function and its derivative.