

Elementary maths

```

ans := expand((a+b)^3):
ans := factor(-2*x+2*x+a*x-x^2+a*x^2-x^3):
ans := {solve(x^2-4 = 0,x)}:
sol := solve(x^2-4 = 0,x):
ans := {x=sol[1],x=sol[2]}:
ans := solve({2*a-b = 3, a+b+c = 1,-b+c = 6},{a,b,c}):
ans := evalf[50](Pi):
ans := convert(1/((1 + x)*(5 + x)),parfrac):
ans := simplify((1/(1 + x) - 1/(5 + x))/4):
ans := simplify(tanh(ln(x))):
ans := simplify(tanh(I*x)):
ans := simplify(sinh(3*x) - 3*sinh(x) - 4*(sinh(x))^3):
ans := ''tanh(ln(x))'':
ans := ''tanh(I*x)'':
ans := ''sinh(3*x) - 3*sinh(x) - 4*(sinh(x))^3'':

```

```

# mpl (ans.101,ans)
# mpl (ans.102,ans)
# mpl (ans.103,ans) {...} avoids maple/latex syntax error
# multiple roots, can't use mpl(foo,bah) here
# mpl (ans.104,ans) fixes problem of multiple roots
# mpl (ans.105,ans)
# mpl (ans.106,ans)
# mpl (ans.107,ans)
# mpl (ans.108,ans)
# mpl (ans.109,ans)
# mpl (ans.110,ans)
# mpl (ans.111,ans)
# mpl (lhs.109,ans)
# mpl (lhs.110,ans)
# mpl (lhs.111,ans)

```

$$\text{ans.101} := a^3 + 3a^2b + 3ab^2 + b^3$$

$$\text{ans.102} := x(x+1)(-x+a)$$

$$\text{ans.103} := \{-2, 2\}$$

$$\text{ans.104} := \{x = -2, x = 2\}$$

$$\text{ans.105} := \left\{ a = \frac{1}{5}, b = -\frac{13}{5}, c = \frac{17}{5} \right\}$$

$$\text{ans.106} := 3.1415926535897932384626433832795028841971693993751$$

$$\text{ans.107} := \frac{1}{4} (x+1)^{-1} - \frac{1}{4} (5+x)^{-1}$$

$$\text{ans.108} := \frac{1}{(x+1)(5+x)}$$

$$\tanh(\ln(x)) = \frac{x^2 - 1}{x^2 + 1} \quad (\text{ans.109})$$

$$\tanh(ix) = i \tan(x) \quad (\text{ans.110})$$

$$0 = \sinh(3x) - 3 \sinh(x) - 4 (\sinh(x))^3 \quad (\text{lhs.111})$$

```

\begin{align*}
&\&\text{mpl}\{ans.101\}\\
&\&\text{mpl}\{ans.102\}\\
&\&\text{mpl}\{ans.103\}\\
&\&\text{mpl}\{ans.104\}\\
&\&\text{mpl}\{ans.105\}\\
&\&\text{mpl}\{ans.106\}\\
&\&\text{mpl}\{ans.107\}\\
&\&\text{mpl}\{ans.108\}\\
&\&\text{lhs.109} \&= \&\text{Mpl}\{ans.109\}\\
&\&\text{lhs.110} \&= \&\text{Mpl}\{ans.110\}\\
&\&\text{lhs.111} \&= \&\text{Mpl}\{lhs.111\}
\end{align*}

```

Linear Algebra

```

with(LinearAlgebra):
mat  := <<2|3>, <5|4>>:           # mpl (ans.201,mat)
ans  := Eigenvectors(mat,output='list'):
eig1 := ans[1][1]:                # 1st eigenvalue
eig2 := ans[2][1]:                # 2nd eigenvalue
v1   := ans[1][3][1]:             # 1st eigenvector
v2   := ans[2][3][1]:             # 2nd eigenvector
eig  := <eig1,eig2>:             # mpl (ans.202,eig)
ans  := <v1|v2>:                  # mpl (ans.203,ans)
ans  := CharacteristicPolynomial(mat,lambda): # mpl (ans.204,ans)
vec  := <3,7>:                    # mpl (ans.205,vec)
sol  := LinearSolve(mat,vec):      # mpl (ans.206,sol)

```

```

\begin{align*}
&\&\text{\texttt{mpl}}*\{\texttt{ans.201}\}\\
&\&\text{\texttt{mpl}}*\{\texttt{ans.202}\}\\
&\&\text{\texttt{mpl}}*\{\texttt{ans.203}\}\\
&\&\text{\texttt{mpl}}*\{\texttt{ans.204}\}\\
&\&\text{\texttt{mpl}}*\{\texttt{ans.205}\}\\
&\&\text{\texttt{mpl}}*\{\texttt{ans.206}\}
\end{align*}

```

$$\begin{aligned}
\texttt{ans.201} &:= \begin{bmatrix} 2 & 3 \\ 5 & 4 \end{bmatrix} \\
\texttt{ans.202} &:= \begin{bmatrix} 7 \\ -1 \end{bmatrix} \\
\texttt{ans.203} &:= \begin{bmatrix} \frac{3}{5} & -1 \\ 1 & 1 \end{bmatrix} \\
\texttt{ans.204} &:= \lambda^2 - 6\lambda - 7 \\
\texttt{ans.205} &:= \begin{bmatrix} 3 \\ 7 \end{bmatrix} \\
\texttt{ans.206} &:= \begin{bmatrix} \frac{9}{7} \\ \frac{1}{7} \end{bmatrix}
\end{aligned}$$

Limits

```
ans := limit(sin(4*x)/x,x=0):           # mpl (ans.301,ans)
ans := limit(2^x/x,x=infinity):         # mpl (ans.302,ans)
ans := limit(((x+dx)^2 - x^2)/dx, dx=0): # mpl (ans.303,ans)
ans := limit((4*n + 1)/(3*n - 1),n=infinity): # mpl (ans.304,ans)
ans := limit((1+(a/n))^n,n=infinity):    # mpl (ans.305,ans)
```

```
\begin{align*}
&\&\text{\texttt{mpl}}\{ans.301\}\\
&\&\text{\texttt{mpl}}\{ans.302\}\\
&\&\text{\texttt{mpl}}\{ans.303\}\\
&\&\text{\texttt{mpl}}\{ans.304\}\\
&\&\text{\texttt{mpl}}\{ans.305\}
\end{align*}
```

```
ans.301 := 4
ans.302 := ∞
ans.303 := 2 x
ans.304 :=  $\frac{4}{3}$ 
ans.305 := ea
```

Series

```
ans := series((1 + x)^(-2), x=1, 6):    # mpl (ans.401,ans)
ans := series(exp(x), x=0, 6):          # mpl (ans.402,ans)
ans := sum(1/n^2, n=1..50):             # mpl (ans.403,ans)
ans := sum(1/n^4, n=1..infinity):       # mpl (ans.404,ans)
```

```
\begin{align*}
&\&\text{\texttt{mpl}}\{ans.401\}\\
&\&\text{\texttt{mpl}}\{ans.402\}\\
&\&\text{\texttt{mpl}}\{ans.403\}\\
&\&\text{\texttt{mpl}}\{ans.404\}
\end{align*}
```

```
ans.401 :=  $\left(\frac{1}{4} - \frac{1}{4}(x-1) + \frac{3}{16}(x-1)^2 - \frac{1}{8}(x-1)^3 + \frac{5}{64}(x-1)^4 - \frac{3}{64}(x-1)^5 + O((x-1)^6)\right)$ 
ans.402 :=  $\left(1 + x + \frac{1}{2}x^2 + \frac{1}{6}x^3 + \frac{1}{24}x^4 + \frac{1}{120}x^5 + O(x^6)\right)$ 
ans.403 :=  $\frac{3121579929551692678469635660835626209661709}{1920815367859463099600511526151929560192000}$ 
ans.404 :=  $\frac{\pi^4}{90}$ 
```

Calculus

```
ans := diff(x*sin(x),x):           # mpl (ans.501,ans)
ans := eval(diff(x*sin(x),x),x=Pi/4): # mpl (ans.502,ans)
ans := int(2*sin(x)^2, x=a..b):     # mpl (ans.503,ans)
ans := int(2*exp(-x^2),x=0..infinity): # mpl (ans.504,ans)
ans := ''int(2*exp(-x^2),x=0..infinity)'' : # mpl (lhs.504,ans)
ans := int(int(x^2 + y^2, y=0..x),x=0..1): # mpl (ans.505,ans)
ans := ''int(int(x^2 + y^2, y=0..x),x=0..1)'' : # mpl (lhs.505,ans)
```

```
\begin{align*}
&\&\text{\texttt{mpl}}*\{\texttt{ans.501}\}\\
&\&\text{\texttt{mpl}}*\{\texttt{ans.502}\}\\
&\&\text{\texttt{mpl}}*\{\texttt{ans.503}\}\\
&\text{\texttt{mpl}}\{\texttt{lhs.504}\}\&=\text{\texttt{Mpl}}\{\texttt{ans.504}\}\\
&\text{\texttt{mpl}}\{\texttt{lhs.505}\}\&=\text{\texttt{Mpl}}\{\texttt{ans.505}\}
\end{align*}
```

$$\texttt{ans.501} := \sin(x) + x \cos(x)$$

$$\texttt{ans.502} := \frac{1}{2} \sqrt{2} + \frac{1}{8} \pi \sqrt{2}$$

$$\texttt{ans.503} := \sin(a) \cos(a) - a - \sin(b) \cos(b) + b$$

$$\int_0^\infty 2e^{-x^2} dx = \sqrt{\pi} \quad (\texttt{ans.504})$$

$$\int_0^1 \int_0^x x^2 + y^2 dy dx = \frac{1}{3} \quad (\texttt{ans.505})$$

Differential equations

```
ode := diff(y(x),x) + y(x) = 2*a*sin(x):
ics := y(0) = 0:
ans := rhs(dsolve(ode)):                # mpl (ans.601,ans)
ans := rhs(dsolve([ics,ode])):          # mpl (ans.602,ans)

ode := diff(y(x),x,x) + y(x) = 0:
ics := y(0)=0, (D(y))(0) = 1:
ans := rhs(dsolve(ode)):                # mpl (ans.603,ans)
ans := rhs(dsolve([ics,ode])):          # mpl (ans.604,ans)

ode := diff(y(x),x,x) + 5*diff(y(x),x) - 6*y(x) = 0:
ans := rhs(dsolve(ode)):                # mpl (ans.605,ans)
sol := eval(ans,[_C1=2,_C2=3]):         # mpl (ans.606,sol)
```

```
\begin{align*}
&\&\text{\texttt{mpl}}\{ans.601\}\\
&\&\text{\texttt{mpl}}\{ans.602\}\\
&\&\text{\texttt{mpl}}\{ans.603\}\\
&\&\text{\texttt{mpl}}\{ans.604\}\\
&\&\text{\texttt{mpl}}\{ans.605\}\\
&\&\text{\texttt{mpl}}\{ans.606\}
\end{align*}
```

$$\text{ans.601} := -a \cos(x) + a \sin(x) + e^{-x} C_1$$

$$\text{ans.602} := -a \cos(x) + a \sin(x) + e^{-x} a$$

$$\text{ans.603} := C_1 \sin(x) + C_2 \cos(x)$$

$$\text{ans.604} := \sin(x)$$

$$\text{ans.605} := C_1 e^x + C_2 e^{-6x}$$

$$\text{ans.606} := 2e^x + 3e^{-6x}$$

A table of derivatives and anti-derivatives

This example is based upon a nice example in the Pythontex gallery, see <https://github.com/gpoore/pythontex/>. It uses a tagged block to capture the Maple output for later use in the body of the LaTeX table.

```
1  # Create a list of functions to include in the table
2  funcs := [[sin(x), "\\\\"], [cos(x), "\\\\"], [tan(x), "\\\\"],
3           [arcsin(x), "\\\\[5pt]"], [arccos(x), "\\\\[5pt]"], [arctan(x), "\\\\[5pt]"],
4           [sinh(x), "\\\\"], [cosh(x), "\\\\"], [tanh(x), " "]]:
5
6  # mplBeg (CalculusTable)
7  for foo in funcs do
8      func := foo[1]:
9      eol   := foo[2]:
10     myddx := ''diff''(func,x):
11     myint  := ''int''(func,x):
12     Print(cat(Latex(myddx), "&=", Latex(diff(func,x)), "\\quad & \\quad")):
13     Print(cat(Latex(myint), "&=", Latex(int(func,x)), eol)):
14 end do:
15 # mplEnd (CalculusTable)
```

```
\begin{align*}
\mpl {CalculusTable}
\end{align*}
```

$$\frac{d}{dx} \sin(x) = \cos(x)$$

$$\frac{d}{dx} \cos(x) = -\sin(x)$$

$$\frac{d}{dx} \tan(x) = 1 + (\tan(x))^2$$

$$\frac{d}{dx} \arcsin(x) = \frac{1}{\sqrt{-x^2 + 1}}$$

$$\frac{d}{dx} \arccos(x) = -\frac{1}{\sqrt{-x^2 + 1}}$$

$$\frac{d}{dx} \arctan(x) = (x^2 + 1)^{-1}$$

$$\frac{d}{dx} \sinh(x) = \cosh(x)$$

$$\frac{d}{dx} \cosh(x) = \sinh(x)$$

$$\frac{d}{dx} \tanh(x) = 1 - (\tanh(x))^2$$

$$\int \sin(x) \, dx = -\cos(x)$$

$$\int \cos(x) \, dx = \sin(x)$$

$$\int \tan(x) \, dx = -\ln(\cos(x))$$

$$\int \arcsin(x) \, dx = x \arcsin(x) + \sqrt{-x^2 + 1}$$

$$\int \arccos(x) \, dx = x \arccos(x) - \sqrt{-x^2 + 1}$$

$$\int \arctan(x) \, dx = x \arctan(x) - \frac{1}{2} \ln(x^2 + 1)$$

$$\int \sinh(x) \, dx = \cosh(x)$$

$$\int \cosh(x) \, dx = \sinh(x)$$

$$\int \tanh(x) \, dx = \ln(\cosh(x))$$

Step-by-step integration

This is another nice example drawn from the Pythontex gallery, see <https://github.com/gpoore/pythontex>.

It shows the step-by-step computations of a simple triple integral.

```
# Define limits of integration
x_max := 2:   y_max := 3:   z_max := 4:
x_min := 0:   y_min := 0:   z_min := 0:

ans := int(f(x,y,z), [x=x_min..x_max, y=y_min..y_max, z=z_min..z_max]):           # mpl(lhs.01,ans)

f := (x,y,z) -> x*y + y*sin(z) + cos(x+y):

ans := ''int''(''int''(''int''(f(x,y,z), x=x_min..x_max), y=y_min..y_max), z=z_min..z_max):   # mpl(rhs.01,ans)
ans := ''int''(''int''(int(f(x,y,z), x=x_min..x_max), y=y_min..y_max), z=z_min..z_max):       # mpl(rhs.02,ans)
ans := ''int''(int(int(f(x,y,z), x=x_min..x_max), y=y_min..y_max), z=z_min..z_max):         # mpl(rhs.03,ans)
ans := int(int(int(f(x,y,z), x=x_min..x_max), y=y_min..y_max), z=z_min..z_max):           # mpl(rhs.04,ans)

# And now, a numerical approximation
ans := evalf[15](ans):                                                                # mpl(rhs.05,ans)
```

$$\begin{aligned}\int_0^4 \int_0^3 \int_0^2 f(x,y,z) \, dx \, dy \, dz &= \int_0^4 \int_0^3 \int_0^2 xy + y \sin(z) + \cos(x+y) \, dx \, dy \, dz \\ &= \int_0^4 \int_0^3 -\sin(y) + 2y + 2y \sin(z) + \sin(2+y) \, dy \, dz \\ &= \int_0^4 8 + \cos(2) + 9 \sin(z) + \cos(3) - \cos(5) \, dz \\ &= 41 + 4 \cos(3) - 4 \cos(5) + 4 \cos(2) - 9 \cos(4) \\ &\approx 40.1235865133292\end{aligned}$$

```
\begin{align*}
\mpl{lhs.01} &= \mpl{rhs.01}\\
&= \mpl{rhs.02}\\
&= \mpl{rhs.03}\\
&= \mpl{rhs.04}\\
&\approx \mpl{rhs.05}
\end{align*}
```


Plotting Bessel functions

This simple example uses Maple to produce a plot of the first six Bessel functions. Two plots are shown, one created by Maple and a second created by LaTeX using the plotting package `pgfplots` and the data exported from Maple.

```
with(plottools):

myPlot := plot([seq(BesselJ(i, z), i = 0 .. 5)], z = 0 .. 15):

exportplot("example-04-fig.jpeg",myPlot,"JPEG"):  # Maple18

a,b,n := 0.0,15.0,150:          # domain and number of samples
dx := (b-a)/n:                  # uniform step

fd := fopen ("example-04.txt", WRITE):
for i from 0 to n by 1 do
  x := a + dx*i:
  fprintf(fd,"% .10e % .10e % .10e % .10e % .10e % .10e % .10e\n",x,
          seq(evalf(BesselJ(k,x)),k=0..5)):
end do:
fclose(fd):
```




Figure 1: The first six Bessel functions.

Using pgfplots




Figure 2: The first six Bessel functions.

```
\begin{tikzpicture} % requires \usepackage{pgfplots}
\begin{axis}
[xmin= 0.0, xmax=15.0,
ymin=-0.45, ymax=1.05,
xlabel=$x$, ylabel=$J_n(x)$,
grid=major, grid style={dashed,gray!30},
legend entries = {$J_0$, $J_1$, $J_2$, $J_3$, $J_4$, $J_5$}]
\addplot[blue] table [x index=0, y index=1]{example-04.txt};
\addplot[red] table [x index=0, y index=2]{example-04.txt};
\addplot[green] table [x index=0, y index=3]{example-04.txt};
\addplot[teal] table [x index=0, y index=4]{example-04.txt};
\addplot[orange] table [x index=0, y index=5]{example-04.txt};
\addplot[purple] table [x index=0, y index=6]{example-04.txt};
\end{axis}
\end{tikzpicture}
\captionof{figure}{The first six Bessel functions.} % requires \usepackage{caption}
```

Displaying long expressions

This example uses a simple (though contrived) example of a Taylor series expansion of $1/(1+x)$ to demonstrate the problems that can arise when displaying very long expressions.

```
f := x -> 1/(1+x):
ans := f(x):           # mpl (ans.511,ans)
ans := series(f(x),x=0, 10): # mpl (ans.512,ans)
ans := series(f(x),x=0, 20): # mpl (ans.513,ans)
ans := series(f(x),x=0, 23): # mpl (ans.514,ans)
```

```
\begin{dgroup*}[spread={5pt}]
\begin{dmath*} f(x) = \Mpl*{ans.511} \end{dmath*}
\begin{dmath*} {}= \Mpl*{ans.512} \end{dmath*}
\begin{dmath*} {}= \Mpl*{ans.513} \end{dmath*}
\begin{dmath*} {}= \Mpl*{ans.514} \end{dmath*}
\begin{dmath*} {}= \Mpl*[\hskip 2cm]{ans.514} \end{dmath*}
\end{dgroup*}
```

The first four lines of the following output were set using `\Mpl*` while the final line used `\Mpl*[\hskip=2cm]`. The last pair of lines displays the output for the same tag `ans.514` and clearly the formatting of the second last line is not ideal as the text has overlapped the tag. This was corrected in the final line by using the option argument `[\hskip=2cm]` in the call to `\Mpl*`.

$$f(x) = (1+x)^{-1} \tag{ans.511}$$

$$= (1-x+x^2-x^3+x^4-x^5+x^6-x^7+x^8-x^9+O(x^{10})) \tag{ans.512}$$

$$= (1-x+x^2-x^3+x^4-x^5+x^6-x^7+x^8-x^9+x^{10}-x^{11}+x^{12}-x^{13}+x^{14}-x^{15}+x^{16}-x^{17}+x^{18}-x^{19}+O(x^{20})) \tag{ans.513}$$

$$= (1-x+x^2-x^3+x^4-x^5+x^6-x^7+x^8-x^9+x^{10}-x^{11}+x^{12}-x^{13}+x^{14}-x^{15}+x^{16}-x^{17}+x^{18}-x^{19}+x^{20}-x^{21}+x^{22}+O(x^{23})) \tag{ans.514}$$

$$= (1-x+x^2-x^3+x^4-x^5+x^6-x^7+x^8-x^9+x^{10}-x^{11}+x^{12}-x^{13}+x^{14}-x^{15}+x^{16}-x^{17}+x^{18}-x^{19}+x^{20}-x^{21}+x^{22}+O(x^{23})) \tag{ans.514}$$

Quadratic convergence of Newton-Raphson iterations

This is a simple example that uses Maple to demonstrate the quadratic convergence of Newton-Raphson iterations to the exact root of a non-linear equation.

```
# mplBeg (table)

Digits := 200:

f := x -> x - exp(-x):
df := x -> eval (diff (f(u),u), u=x):

x_new := 0.5:
f_new := f (x_new):

Print (sprintf("\RuleA % 2d &% .25f &% .10e &\\",
               0,x_new,f_new)):

for n from 1 to 6 do
  x_old := x_new:
  x_new := x_old - f(x_old)/df(x_old):
  f_old := evalf (f (x_old)):
  f_new := evalf (f (x_new)):
  ratio := evalf (f_new / f_old^2):
  Print (sprintf("\RuleA % 2d &% .25f &% .10e &% .5f\\",
                 n,x_new,f_new,ratio)):
end do:

# mplEnd (table)
```

```
\def\eps{\epsilon}
\def\RuleA{\vrule depth0pt width0pt height14pt}
\def\RuleB{\vrule depth8pt width0pt height14pt}
\def\RuleC{\vrule depth10pt width0pt height16pt}

\setlength{\tabcolsep}{0.025\textwidth}%

\begin{center}
\begin{tabular}{cccc}%
  \noalign{\hrule height 1pt}
  \multicolumn{4}{c}{\RuleC\rmfamily\bfseries%
    Newton-Raphson iterations \quad%
    $x_{n+1} = x_n - f_n/f'_n$, \quad $f(x) = x-e^{-x}$}\\
  \noalign{\hrule height 1pt}
  \RuleB$n$&$x_n$&
    $\eps_n = x_n - e^{-x_n}$&
    $\eps_n/\eps_{n-1}^2$\\
  \noalign{\hrule height 0.5pt}
  \mpl{table}
  \noalign{\hrule height 1pt}
\end{tabular}
\end{center}
```

Note the clear quadratic convergence in the iterations – the last column settles to approximately -0.11546 independent of the number of iterations. This behaviour would not be seen using normal floating point computations as they are normally limited to no more than 18 decimal digits. This computation used 200 decimal digits.

Newton-Raphson iterations $x_{n+1} = x_n - f_n/f'_n$, $f(x) = x - e^{-x}$			
n	x_n	$\epsilon_n = x_n - e^{-x_n}$	$\epsilon_n/\epsilon_{n-1}^2$
0	0.500000000000000000000000	-1.0653065971e-01	
1	0.5663110031972181530416492	-1.3045098060e-03	-0.11495
2	0.5671431650348622127865121	-1.9648047172e-07	-0.11546
3	0.5671432904097810286995766	-4.4574262753e-15	-0.11546
4	0.5671432904097838729999687	-2.2941072910e-30	-0.11546
5	0.5671432904097838729999687	-6.0767705445e-61	-0.11546
6	0.5671432904097838729999687	-4.2637434326e-122	-0.11546

Using tagged blocks

The following Maple code block contains a matched `mplBeg/mplEnd` pair, with the tag name `info`, to capture the output from the formatted Maple `Print` statements.

```
# mplBeg(info)
Print(cat("date :  &",FormatTime("%a %d %b %Y %H:%M:%S"),"\\\\\\"):
Print(cat("pwd :   &",getenv("PWD"),"\\\\\\"):
Print(cat("shell : &",getenv("SHELL"),"\\\\\\"):
Print(cat("home :  &",getenv("HOME"))):
# mplEnd(info)
```

```
\bgroup\tt
\begin{tabular}{rl}
  \mpl{info}
\end{tabular}
\egroup
```

Here is the output caught from the above block.

```
date :  Mon 27 Aug 2018 10:16:04
pwd :   /Users/leo/GitHub/leo-brewin/hybrid-latex/maple/examples
shell :  /usr/local/bin/bash
home :   /Users/leo
```

A mixed Maple-Python example

This example demonstrates a cooperative effort where Maple is used to do the analytic computations while Python is used to plot the data.

The example chosen here is to find and plot the solution to the boundary value problem defined by

$$\frac{d^2y}{dx^2} + 2\frac{dy}{dx} + 10y = 0 \quad \text{with } y(0) = 3, y'(0) = 0$$

This example requires two passes, once for Maple and once for Python (and in that order). This example can be run using

```
mpllatex.sh -x -i mixed
pylatex.sh -x -i mixed
pdflatex      mixed
```

Note that the last pair of commands could also be combined as `pylatex.sh -i mixed`.

The Maple code

Here Maple is used to first find the general solution of the differential equation. The boundary conditions are then imposed and finally a uniform sampling of the solution is written to a file for later use by Python and Matplotlib.

```
# a second order ode
ode := diff(y(x), x, x) + 2*diff(y(x), x) + 10*y(x) = 0: # mpl (ans.101,ode)

# find the general solution
ans := dsolve(ode): # mpl (ans.102,ans)

# set initial conditions
ics := y(0) = 3, (D(y))(0) = 0:
tmp := {ics}: # mpl (ans.103,tmp)

# find the particular solution
f := rhs(dsolve([ics, ode])): # mpl (ans.104,f)
df := diff(f,x): # mpl (ans.105,df)

y := x -> f:
dy := x -> df:

# now sample y and dy at selected points
a,b,n := 0.0,2.0*Pi,300: # domain and number of samples
dx := (b-a)/n: # uniform step

fd := fopen ("mixed.txt", WRITE):
for i from 0 to n by 1 do
    x := a + dx*i:
    fprintf(fd,"% .10e % .10e % .10e\n",x,evalf(y(x)),evalf(dy(x))):
end do:
fclose(fd):
```


The general solution of the differential equation is

$$y(x) = C_1 e^{-x} \sin(3x) + C_2 e^{-x} \cos(3x)$$

while the particular solution satisfying the boundary conditions is given by

$$y(x) = e^{-x} \sin(3x) + 3e^{-x} \cos(3x)$$

The Python code

This is a straightforward use of Matplotlib to plot two functions. The code reads the datafile created previously by Maple and then calls Matplotlib to plot that data.

```
import numpy as np
import matplotlib.pyplot as plt

plt.matplotlib.rc('text', usetex = True)
plt.matplotlib.rc('grid', linestyle = 'dotted')
plt.matplotlib.rc('figure', figsize = (5.5,4.1)) # (width,height) inches

x, y, dy = np.loadtxt('mixed.txt', unpack=True)

plt.plot(x,y)
plt.plot(x,dy)

plt.xlim(0.0,4.0)

plt.legend(('y(x)', 'dy(x)/dx'), loc = 0)
plt.xlabel('$x$')
plt.ylabel('$y(x), \> dy/dx$')
plt.grid(True)
plt.tight_layout(0.5)

plt.savefig('mixed-fig.pdf')
```




Figure 1: The function and its derivative.